

SIMPL

MSX Computer Magazine nummer 50 - december 1991

Scanned, ocr'ed and converted to PDF by HansO, 2001


MSE: MCM's Sample Editor SIMPL heet de hardware, en dat is 'ie ook. Met een half uurtje werk kan iedereen voortaan gedigitaliseerd geluid afspelen, op zowat elke MSX. De enige eis: 32 kB RAM-geheugen. MSE, hier beschreven, is de eerste toepassing van ons feest-project. Na 49 nummers gaat MCM op de muzikale toer.

MSX Computer Magazine viert haar vijftigste nummer. En dat heeft de redactie geweten ook. Zo is dit bouwproject - waarmee ook de nederige MSX1 opeens echt gesampled geluid kan weergeven - onder de knoet van de hoofdredacteur uitgewerkt. Een leuk ideeetje, dat opeens haast kreeg. Want MCM 50 is een geluidsspecial. En dit bouwproject is één van onze cadeautjes aan onze lezers. We zijn er best trots op...

Computers worden in de hedendaagse muziek veelal gebruikt voor het afspelen en bewerken van zogenaamde samples. Deze samples ontstaan nadat een geluidsfragment - bijvoorbeeld een drumgeluid, het geluid van een viool of om maar wat té noemen, een kwakende eend - is omgezet naar voor een computer begrijpelijk signaal. Het afspelen van een sample gebeurt met een D/A-converter. Deze zet een digitaal computersignaal om in een voor een mens hoorbaar - analoog - signaal. Ook de MSX turbo R, de MSX Audio en de Philips Music Module maken gebruik van deze techniek. We zullen u nu voorstellen aan SIMPL, een D/A-converter die elke MSX omtovert tot een elektronische muziekdoo die zijn weerga niet kent.

SIMPL - kort voor Sample Interface MSX PLayer - kan samples afspelen via de printerpoort. Met dit stukje low-cost hardware kunnen diverse nieuwe toepassingen worden gerealiseerd op de MSX, zoals een sprekende klok of een drumcomputer. Maar ook het afspelen van turbo R samples is nu op elke MSX - ook MSX1 - mogelijk. Elders in dit blad vindt u een eerste voorbeeld: MCM's Sample Editor, een programma om samples mee te editten, knippen, plakken en filteren.

SIMPL wordt aangesloten op de acht datalijnen van de printer-aansluiting. De elektronica zet vervolgens elke ontvangen waarde om naar een overeenkomstige spanning. Die hoorbaar gemaakt kan worden via de monitor of een versterker.



Schema SMPL D/A converter

Onderdelenlijst

- 9 metaalfilm-weerstanden van 11 kilo Ohm (kleurcode bruin,bruin,zwart,rood)
- 9 metaalfilm-weerstanden van 22 kilo Ohm (kleurcode rood,rood,zwart,rood)
- 1 condensator van 100 nano Farad
- 1 printerstekker (14 pins centronics male)
- een stukje flatcable (9 aderig)
- 2 tulpstekers

Het schema

Het SIMPL-schema is een zogenaamd R/2R-netwerk. Hiermee wordt de binaire data die de computer naar de printerpoort stuurt omgezet naar een analogo signaal. Het beste is om zogenaamde metaalfilm-weerstanden te gebruiken; dit zijn weerstanden met een tolerantie van slechts 1%, waardoor SIMPL geen ruis produceert.

Maar ook gewone weerstanden kunnen prima voldoen. Voor het bouwen van SIMPL kan men het makkelijkst een experimenteer printplaatje gebruiken. Dit is een printplaatje met een aantal rijen gaatjes, waarop de elektronische componenten gesoldeerd worden.

De monitor of versterker wordt uiteindelijk via een tulpstekertje verbonden met SIMPL. Sommige versterkers kunnen niet in een monostand geschakeld worden, waardoor het geluid slechts door één luidspreker hoorbaar is. Dit is op te lossen door dan twee tulpstekertje te gebruiken. Zie het schema voor de juiste aansluiting van de tweede tulpstekker.

Machines met een 7 MHz uitbreiding moeten in de normale stand geschakeld worden. Door de hogere snelheid namelijk worden de samples te snel afgespeeld. De turbo R bezitter heeft geluk; bij deze computer is standaard PCM - Puls Code Modulatie - ingebouwd zodat SIMPL overbodig is.

Overigens kunnen op de turbo R de samples 'slechts' in vier verschillende snelheden worden weergegeven. SIMPL zelf kent deze beperking niet en kan dus een sample op veel verschillende snelheden afspelen. De snelheid van het afspelen bepaalt de frequentie en dus de uiteindelijke toonhoogte waarop de sample ten gehore wordt gebracht.

Waar verkrijgbaar

Op zich zijn alle onderdelen voor de Sample Interface MSX PLayer, kortweg SIMPL genaamd, in elke goed gesorteerde elektronika-onderdelen winkel te koop. Maar achter de toonbank van zo'n zaak staat meestal een meneer in een stofjas, die er van uit wenst te gaan dat zijn klanten toch dom zijn. Als men het jargon niet kent, dan komt men vaak met een kater en soms de verkeerde onderdelen thuis. Wie de weg en het taaltje kent, die kan in dat soort zaken terecht.


Samples

De bekende MCM programmaservice diskette bij dit nummer, MD 49, bevat naast de MCM's Sample Editor – zie elders dit nummer - ook diverse samples. Maar we hopen eigenlijk dat SIMPL een rage zal worden, en dat iedereen die daar kans toe ziet samples gaat verspreiden.


Samen met de Sample Editor elders in ons feestnummer - had u al gemerkt dat dit de geluids-special is? - kan iedereen voortaan spelen met geluid. Zonder FM-Pac, Music Module, MSX2+ of wat dan ook: gewoon op een MSX1 of hoger.

MSE - ofwel MCM's Sample Editor - is een eerste toepassing voor SIMPL, een D/A converter welke u voor weinig geld zelf kunt bouwen. MSE werkt op alle MSX computers, dus ook op MSX1, terwijl het programma op de MSX turbo R gebruik maakt

van de aanwezige PCM-chip. MCM's Sample Editor kan samples filteren, grafisch weergeven en afspelen met behulp van SIMPL. Ook turbo R samples kunnen door MSE verwerkt worden. Maar voordat we u hier meer over vertellen, eerst een vrij technisch verhaal over sampling. Desgewenst mag u dit overslaan; ook zonder de technische details zijn samples leuk speelgoed.



Figuur 1: Bemonstering van een analoog signaal



Figuur 2: Resultaat van bemonstering figuur 1

Sampling

Sampling is een techniek — welke onder andere bij het maken van CD's wordt toegepast - waarbij geluiden omgezet worden naar een digitaal signaal. Zo'n digitaal signaal, oftewel: gewoon data, kunnen we bijvoorbeeld opslaan in het geheugen van een computer of een synthesizer. Maar eerst moet het geluid -een analoog signaal - worden geconverteerd.

In figuur 1 zien we een klein fragment van een analoog signaal grafisch weergegeven. De horizontale as stelt de tijd voor, op de verticale as staat de amplitude - de 'hoogte' op een bepaald tijdstip - van de golf weergegeven. Zo is de amplitude van de sinusgolf op tijdstip twee gelijk aan vier en minus twee op tijdstip vijf.

Om een analoog signaal om te zetten naar digitale data dienen we het signaal te vast te leggen in getallen. Dat heet bemonsteren: de amplitude van het signaal op vaste tijdstippen omzetten naar een digitale waarde. In ons voorbeeld krijgen we dan de reeks:

0, 2,4 ,3, 0, -2, -4, -4, -2. Zoals we zien worden de waarden afgerond naar een geheel getal. Dit wordt het kwantiseren - spreek uit: kwantiseren - van de sample genoemd.

Een D/A - Digitaal - AnalooG - conver-ter, zoals SIMPL, kan deze getallen later weer omzetten naar analooG geluid. Het resultaat is weergegeven in figuur 2. Zoals we al kunnen zien wijkt dit nogal af van het oorspronkelijke analooG signaal van figuur 1. De uiteindelijke kwaliteit van het geluid na D/A conversie is afhankelijk van twee belangrijke factoren, namelijk sample-frequentie en de sample-resolutie.

Sample-frequentie

Ten eerste moet de sample-frequentie -het aantal metingen per seconde - zo hoog mogelijk zijn. Als het analooG signaal vaker wordt bemonsterd krijgen we een beter eindresultaat. In figuur 1 wordt een zeer lage sample-frequentie gebruikt. Tussen twee willekeurige tijdstippen kunnen we het signaal rustig wijzigen -zie de stippelijN tussen tijdstip drie en vier - zonder dat het resultaat in figuur 2 er door zal veranderen. Als we bijvoorbeeld een geluid 'samplen' met een snelheid van 20 kHz - dit is 20000 maal per seconde - dan is de tijd tussen twee metingen slechts 0,00005 seconde.

Resolutie

Een tweede belangrijke factor die het eindresultaatbeïnvloedt is de zogenaamde sample-resolutie. Dit is het aantal stapjes waarin we de verticale as verdelen. Hoe meer stapjes, hoe hoger de resolutie en des te beter het eindresultaat. In ons voorbeeld - zie figuur 1 - is de verticale as slechts verdeeld in acht stappen van -4 tot 4, ofwel 9 verschillende toestanden. Figuur 3 toont het gedigitaliseerde signaal van figuur 1, echter nu met een hogere sample-frequentie en resolutie, waardoor het oorspronkelijke signaal veel beter benaderd wordt.

SIMPL kan samples afspelen met een 8 bits resolutie, dit zijn 256 stapjes lopend van -128 tot 127. Acht bits samples worden veel gebruikt, zoals onder andere op de MSX turbo R. Voor een redelijk resultaat is minimaal een 8 bits resolutie noodzakelijk en een sample-frequentie van ongeveer 16 kHz. CD-spelers werken met een 16 bits sample-resolutie en een sample-frequentie van 44.1 kHz, wat resulteert in een perfecte weergave.

De lol van digitaal

Voor de kwaliteit hoeven we met een MSX geen samples te gaan maken. Een beetje cassetterecorder doet het stukken beter en vooral, veel langer. Want samples vreten letterlijk geheugen, met enkele seconden digitaal geluid is de grens voor een MSX-geheugen meestal wel bereikt.


Maar als een geluid eenmaal in het geheugen van de computer staat, dan kunnen we daar allerlei bewerkingen op los laten. Bijvoorbeeld het optellen van twee samples, het filteren van ruis of het vervormen van de klank. Het grote voordeel van bewerkingen op een digitaal signaal, is dat we geen last hebben van kwaliteitsverlies: het is gewoon rekenen. Als voorbeelden omvat MCM's Sample Editor een echo filter en een zogenaamd low pass filter, waarmee hoge tonen uit een sample worden weggefilterd.

Ook aardig: we kunnen een sample makkelijk grafisch weergeven - zie figuur 4 - zodat we een indruk krijgen hoe het gesampelde geluid er grofweg uitziet. In figuur 4 zien we links duidelijk een andere sample dan rechts. Het linkergedeelte is de grafische weergave van een bassdrum. Duidelijk is de aanslag - of attack - van de bassdrum te zien. De rechter grafiek is die van een snaredrum. Opvallend hierbij is de hoeveelheid ruis in de sample. Dit is te zien aan het grote aantal pieken in de grafiek.


MSE

MCM's Sample Editor is voor hetgrootste deel in Basic geschreven. Een klein machinetaalgedeelte verzorgt het filteren, het afspelen van de samples en nog een aantal zaken waar Basic nu eenmaal te langzaam voor is.

Het afspelen van samples gebeurt door het sturen van waarden naar de printerpoort -I/O poort-adres &h91 - waarop SIMPL, de in onze rubriek de Soldeerbout omschreven zelfbouw Sample Interface MSX PLayer, is aangesloten. Door de wachttijd tussen deze waarden te variëren zal het geluid in toonhoogte veranderen. Met de speed-optie — functietoets vier — kunnen we die wachttijd veranderen en dus spelen met de toonhoogte.



Figuur 3: Resultaat van bemonstering met hogere sample-frequentie en resolutie



Figuur 4. Grafische weergave van een sample

Filters

Zoals al eerder vermeld, bevat MCM's Sample Editor twee verschillende filters om de sample te wijzigen. Met de low pass filter - functietoets F2 - kunnen de lage tonen versterkt worden. Het meerdere malen activeren van deze optie heeft tot gevolg dat het geluid steeds lager gaat klinken. Een aardige bijkomstigheid is dat ook de ruis wordt weggefilterd. De tweede filter - activeren met functietoets F3 - kan gebruikt worden om echo aan een sample toe te voegen. Meerdere malen een sample van echo voorzien resulteert in een blikkerig 'robot' geluid.

De Select toets kan worden gebruikt om samples van disk of cassette te laden of te save. Bij diskette wordt de extensie .SMP gebruikt. Deze bestanden zijn 16 kB - ofwel één segment - groot.

Samples

Op de MCM's ProgrammaService diskette bij dit nummer - MCM 50, dus disk 49 — staan een aantal voorbeeld-samples die u kunt inladen bij MSE. Hoe we die gemaakt hebben is geen groot geheim: de turbo R kan niet alleen samples afspelen, maar ook opnemen. Hiertoe is deze computer van een aantal Basic uitbreidingen voorzien. SAMPTR, één van de listings bij dit artikel, is een kort Basic programma dat deze mogelijkheden demonstreert. Met SAMPTR gemaakte samples kunnen met de extensie SMP worden weggeschreven en vervolgens bij MSE worden ingeladen. En wie het weten "wil: SAMPTR staat voor SAMPples op de Turbo R.

De listing van MCM's Sample Editor bestaat uit twee gedeelten, een loader die naast het installeren van Basic in page 3, het machinetaal gedeelte aanmaakt. Disk gebruikers dienen deze loader weg te schrijven onder de naam 'MSE.LDR', cassette gebruikers kunnen deze loader het beste onder de naam 'MSELDR' wegschrijven. De tweede listing bevat het eigenlijke Basic programma - wegschrijven onder de naam 'MSE.BAS', cassette gebruikers kunnen hier de extensie weer weglaten.

Andere programma's

De programma's die je nodig hebt om SIMPL te laten werken, er zijn er vele, bv PSG-SAMPLER.

Dit programma is te gebruiken omdat je hiermee zelf gemakkelijk samples kan maken via je MUSIC-MODULE en op allerlei andere music chips kan afspelen. Het programma is ook erg makkelijk te gebruiken.

Je kan eventueel ook nog wat sample'tjes inladen en dan beluisteren in PSG-SAMPLER.

```

10 REM MSE Loader
20 REM
30 REM
40 REM
50 REM MSX Computer Magazine
60 REM
70 REM
80 CLS: PRINT"MCM's Sample Editor": PRINT
90 PRINT "Installeren machinetaal..."
100 RESTORE 190: AD=&HC000: CK=0
110 READ A$: IF A$="*" THEN 140
120 A=VAL("&h"+A$): POKE AD,A: AD=AD+1: CK=CK+A
130 GOTO 110
140 IF CK<>42125! THEN PRINT "Fout in dataregels!": END
150 PRINT "MSE wordt nu geladen..."
160 POKE &HF676,1: POKE &HF677,&HC2: POKE &HC200,0
170 IF PEEK (&HFFA7)<>201 THEN RUN "MSE.BAS"
180 RUN "MSE"
190 DATA 18,1B,18,51,00,FF,C3,0D
200 DATA C1,07,00,00,00,00,00,00
210 DATA C3,FC,C0,00,00,00,00,C3
220 DATA 45,C1,C3,68,C1,F6,FF,32
230 DATA 05,C0,AF,32,04,C0,CD,84
240 DATA 00,11,A4,C0,06,10,1A,13
250 DATA CD,74,C0,23,10,F8,06,10
260 DATA AF,CD,74,C0,23,10,F9,AF
270 DATA CD,87,00,11,BC,C0,06,40
280 DATA 1A,13,CD,74,C0,23,10,F8
290 DATA CD,55,C0,FB,C9,AF,CD,87
300 DATA 00,23,3A,04,C0,11,04,00
310 DATA CD,6A,C0,3A,05,C0,CD,6A
320 DATA C0,C9,06,08,CD,74,C0,19
330 DATA 10,FA,FB,C9,F5,3A,AF,FC
340 DATA FE,02,28,07,CD,88,C0,F1
350 DATA ED,79,C9,F1,CD,4D,00,C9
360 DATA 3A,07,00,F3,3C,4F,7C,E6
370 DATA C0,07,07,ED,79,3E,8E,ED
380 DATA 79,7C,ED,69,E6,3F,F6,40
390 DATA ED,79,0D,C9,80,80,00,00
400 DATA 80,80,00,00,80,80,00,00
410 DATA 80,80,00,00,80,80,00,00
420 DATA 80,80,00,00,00,00,00,0E
430 DATA 10,00,00,0E,20,00,00,0E
440 DATA 30,00,00,0E,40,00,00,0E
450 DATA 50,00,00,0E,60,00,00,0E
460 DATA 70,00,00,0E,00,FF,00,0E
470 DATA 10,FF,00,0E,20,FF,00,0E
480 DATA 30,FF,00,0E,40,FF,00,0E
490 DATA 50,FF,00,0E,60,FF,00,0E
500 DATA 70,FF,00,0E,F3,2A,13,C0
510 DATA 3A,2D,00,FE,03,28,27,ED
520 DATA 5B,15,C0,18,0F,F3,2A,0C
530 DATA C0,3A,2D,00,FE,03,28,1C
540 DATA ED,5B,0E,C0,7E,D3,91,23
550 DATA 1B,3A,09,C0,47,10,FE,7A

```


560 DATA B3,C2,1C,C1,FB,C9,ED,4B
570 DATA 15,C0,18,04,ED,4B,0E,C0
580 DATA 11,00,00,3A,0A,C0,E6,03
590 DATA CD,86,01,FB,C9,F3,2A,0C
600 DATA C0,ED,5B,0E,C0,1B,7E,D6
610 DATA 80,47,23,7E,D6,80,90,CB
620 DATA 2F,80,C6,80,77,D6,80,47
630 DATA 1B,7A,B3,C2,52,C1,FB,C9
640 DATA F3,01,00,01,2A,0E,C0,ED
650 DATA 42,EB,1B,1B,2A,0C,C0,01
660 DATA 00,01,09,7E,D6,80,CB,2F
670 DATA E5,C5,ED,42,47,7E,D6,80
680 DATA CB,2F,80,C6,80,C1,E1,77
690 DATA 23,1B,7A,B3,C2,7B,C1,FB
700 DATA C9,*

```

10 REM MSE
20 REM
30 REM MCM's Sample Editor
40 REM
50 REM MSX Computer Magazine
60 REM
70 REM
80 IF PEEK(&HF677)<>&HC2 THEN PRINT "Dit programma starten met MSE
Loader": END
90 DEFINT C-Z: M=PEEK(&H2D): T=0
100 IF M=3 THEN T=1
110 IF M>0 THEN M=1
120 XM=255+256*M: YM=191+20*M: L1=0: L2=255
130 AS=&H8000: AE=&HBFFF: SP=0: SI=10: TR=0: C2=0
140 DIM K(9,2)
150 DEFUSR0=&HC000: DEFUSR1=&HC002: DEFUSR2=&HC006
160 DEFUSR3=&HC010: DEFUSR4=&HC017: DEFUSR5=&HC01A
170 OPEN "GRP:" FOR OUTPUT AS #1
180 COLOR 15,0,0: SCREEN 2+M*5,2
190 GOSUB 440: GOSUB 330
200 ' hoofdprogramma *****
210 ON KEY GOSUB 950,1050,1080,1370,1110
220 FOR I=1 TO 5: KEY(I)ON: NEXT I
230 U$=INKEY$: U=VAL(U$)
240 IF U>=1 AND U<=9 THEN 1010
250 IF U$=CHR$(13) THEN GOSUB 900
260 IF U$=CHR$(24) THEN GOTO 1150
270 IF PEEK(&HFBE) AND 1 THEN SH=1 ELSE SH=6
280 C=STICK(0): IF C=0 OR (C MOD 2)=0 THEN GOTO 310
290 IF C=3 THEN GOTO 650 ELSE IF C=7 THEN GOTO 590
300 SP=C/5
310 IF STRIG(0) THEN R=&HC00C: GOSUB 810: A=USR2(0)
320 GOTO 230
330 ' bereken + grafische weergave **
340 GOSUB 1500
350 B1=(AE-AS)/(L2-L1): B2=B1/(M+1)
360 H=AS
370 FOR X=L1*(M+1) TO L2*(M+1)
380   LINE (X,0)-(X,129),0
390   LINE (X,65)-(X,65-(PEEK(H)-128)/2),6
400   H=H+B2
410 NEXT X
420 GOSUB 1470
430 RETURN
440 ' scherm opbouw *****
450 LINE (0,131)-(XM,YM),4,B
460 A=USR0(0): COLOR 15
470 PRESET (8,140): PRINT #1,"MCM's Sample Editor"
480 PRESET (8,YM-20): PRINT #1,"Select=Load/Save Return=zoom"
490 PRESET (8,YM-10): PRINT #1,"store lpf echo speed unzoom"
500 GOSUB 530: GOSUB 560
510 COLOR 3: A=USR0(0)
520 RETURN
530 ' plaats sample pointer 1 *****
540 POKE &HC004,L1: A=USR1(0)
550 RETURN

```

```

560 ' plaats sample pointer 2 *****
570 POKE &HC005,L2: A=USR1(0)
580 RETURN
590 ' pointer 1 naar links *****
600 IF SP THEN GOTO 720
610 IF L1<SH THEN GOTO 230
620 L1=L1-SH: GOSUB 530
630 AS=AS-B1*SH
640 GOTO 230
650 ' pointer 1 naar rechts *****
660 IF SP THEN GOTO 770
670 IF L1>=L2-SH THEN GOTO 230
680 L1=L1+SH: GOSUB 530
690 AS=AS+B1*SH
700 GOTO 230
710 ' pointer 2 naar links *****
720 IF L2<=L1+SH THEN GOTO 230
730 L2=L2-SH: GOSUB 560
740 AE=AE-B1*SH
750 GOTO 230
760 ' pointer 2 naar rechts *****
770 IF L2=>256-SH THEN GOTO 230
780 L2=L2+SH: GOSUB 560
790 AE=AE+B1*SH
800 GOTO 230
810 ' init start + lengte sample ****
820 S2=AS: AD=VARPTR(S2)
830 POKE R,PEEK(AD)
840 POKE R+1,PEEK(AD+1)
850 E2=AE-AS: AD=VARPTR(E2)
860 POKE R+2,PEEK(AD)
870 POKE R+3,PEEK(AD+1)
880 POKE &HC009,SI: POKE &HC00A,TR
890 RETURN
900 ' zoom *****
910 IF AE-AS<XM THEN RETURN
920 L1=0: L2=255
930 GOSUB 1500: GOSUB 540: GOSUB 560: GOSUB 330: GOSUB 1470
940 RETURN
950 ' store *****
960 PRESET (8,154): PRINT #1,"store :";
970 A$=INPUT$(1): PRINT #1,A$
980 IF VAL(A$)<1 OR VAL(A$)>9 THEN GOSUB 1470: GOTO 960
990 K(VAL(A$),1)=AS: K(VAL(A$),2)=AE: GOSUB 1470
1000 RETURN
1010 ' play sample onder toets *****
1020 I=AS: J=AE: AS=K(U,1): AE=K(U,2)
1030 R=&HC013: GOSUB 810: AS=I: AE=J
1040 A=USR3(0): GOTO 230
1050 ' low pass filter *****
1060 GOSUB 1500: R=&HC00C: GOSUB 810: A=USR4(0): GOSUB 330: GOSUB 1470
1070 RETURN
1080 ' echo *****
1090 GOSUB 1500: R=&HC00C: GOSUB 810: A=USR5(0): GOSUB 330: GOSUB 1470
1100 RETURN
1110 ' unzoom *****
1120 AS=&H8000: AE=&HBFFF: L1=0: L2=255

```

```

1130 GOSUB 1500: GOSUB 530: GOSUB 560: GOSUB 330: GOSUB 1470
1140 RETURN
1150 ' I/O menu *****
1160 ON ERROR GOTO 1320
1170 COLOR 15,4,4: SCREEN 0
1180 PRINT "*** Samples laden en saven ***": PRINT
1190 PRINT "1. Load disk": PRINT "2. Save disk"
1200 PRINT "3. Load cassette": PRINT"4. Save cassette"
1210 PRINT: PRINT "Uw keuze?";: A$=INPUT$(1)
1220 K=VAL(A$): PRINT K: PRINT
1230 IF K<1 OR K>4 THEN GOTO 1170
1240 IF K<3 THEN F$="": FILES F$+"*.SMP" ELSE F$="CAS:"
1250 PRINT: PRINT
1260 A$="": INPUT "Filenaam (zonder extensie)";A$
1270 PRINT "momentje..."
1280 IF K<3 THEN A$=A$+".SMP"
1290 IF K=1 OR K=3 THEN BLOAD F$+A$ ELSE BSAVE F$+A$,AS,AE
1300 ON ERROR GOTO 0
1310 GOTO 180
1320 IF ERR=56 THEN PRINT"Bestandsnaam is onjuist!": GOTO 1350
1330 IF ERR=53 THEN PRINT"Sample niet gevonden!": GOTO 1350
1340 PRINT "I/O error"
1350 PRINT "Druk toets voor vervolg": A$=INPUT$(1)
1360 RESUME 1170
1370 ' afspeelsnelheid *****
1380 PRESET (8,154)
1390 IF PEEK(&H2D)<3 THEN GOTO 1440
1400 PRINT#1,"Turbo R scale rate (";TR;):";
1410 A$=INPUT$(1): PRINT #1,A$: IF A$="" THEN GOTO 1430
1420 IF A$<"0" OR A$>"3" THEN GOTO 1410 ELSE TR=VAL(A$)
1430 PRESET (8,154): PRINT #1,SPACE$(26): RETURN
1440 PRINT #1,"Speed level (";SI;):";: A=0: T=2
1450 A$=INPUT$(1): PRINT #1,A$;: IF A$=CHR$(13) THEN GOTO 1460 ELSE
A=A*10+VAL(A$): T=T-1: IF T=0 THEN GOTO 1460 ELSE GOTO 1450
1460 IF A<1 OR A>99 THEN 1480 ELSE SI=A
1470 ' maak invoer regel schoon *****
1480 LINE (8,154)-(255,162),1,BF
1490 RETURN
1500 ' even wachten *****
1510 PRESET (8,154): PRINT #1,"momentje..."
1520 RETURN

```

```
10 REM SAMPTR
20 REM
30 REM wegschrijven als "SAMPTR.BAS"
40 REM
50 REM MSX Computer Magazine
60 REM
70 REM Alleen MSX turbo R, R800 mode!
80 REM
90 IF PEEK (&HF677)<>&HC0 THEN POKE &HF677,&HC0: POKE &HC000,0: RUN
"SAMPTR.BAS"
100 CLS: PRINT "Sampling op MSX turbo R"
110 PRINT
120 PRINT "[1] afspelen sample"
130 PRINT "[2] opnemen sample"
140 PRINT "[3] wegschrijven naar disk"
150 PRINT: PRINT "Uw keuze?";: A$=INPUT$(1)
160 K=VAL(A$)
170 IF K<1 OR K>3 THEN GOTO 110
180 ON K GOTO 190,250,200
190 CALL PCMPPLAY (@&H8000,&HBFFF,0): GOTO 110
200 PRINT: PRINT "filenaam (zonder extensie)";: INPUT F$
210 F$=LEFT$(F$,8)+".smp"
220 PRINT "Bezig met wegschrijven..."
230 BSAVE F$,&H8000,&HBFFF
240 GOTO 110
250 PRINT: PRINT "Druk toets op te starten met opnemen...": A$=INPUT$(1)
260 PRINT "Bezig met opnemen..."
270 CALL PCMREC (@&H8000,&HBFFF,0,127)
280 GOTO 110
```